Online supplementary file for „An Optimized Hierarchical Two-Parameter Logistic Model for Small-Sample Item Calibration“: *Rstan* code of the optimized hierarchical two-parameter logistic model.

```
optim_H2PL <- "

data {                                  // # data specification
  int<lower=1> I;                       // items
  int<lower=1> J;                       // persons
  int<lower=1> N;                       // observations (responses)
  int<lower=1, upper=I> ii[N];          // item for n
  int<lower=1, upper=J> jj[N];          // person for n
  int<lower=0, upper=1> y[N];           // correctness for n
}

parameters {                            // # parameter specification
  vector[J] theta;                      // abilities
  matrix[2,I] xi_tilde;                 // z-score item parameters (Eq. 2.3)
  vector[2] mu;                         // item parameter grand means
  vector<lower=0>[2] tau;               // item parameter variance components
  cholesky_factor_corr[2] L_Omega;      // Cholesky factor of Σ

}

transformed parameters {                // # parameter transformations
  matrix[I,2] xi;                       // log_alpha, beta
  vector<lower=0>[I] alpha;             // item discrimination
  vector[I] beta;                       // item difficulty

  xi = (diag_pre_multiply(tau, L_Omega) * xi_tilde)';   // Transformation 1

  for (i in 1:I) {                      // Transformation 2:
   alpha[i] = exp(mu[1] + xi[i, 1]);    // Glas & van der Linden, 2003
   beta[i] = mu[2] + xi[i, 2];
  }
}

model {                                 // # model specification

  theta ~ normal(0,1);                  // Eq. 2.2
  to_vector(xi_tilde) ~ normal(0,1);    // Eq. 2.3 (non-centering)

  mu[1] ~ normal(0,1);                  // Eq. 2.4
  mu[2] ~ normal(0,2);                  // Eq. 2.5

  L_Omega ~ lkj_corr_cholesky(4);       // Eq. 2.6 (separation strategy)
  tau ~ cauchy(0,1);                    // Eq. 2.7 (separation strategy)

  y ~ bernoulli_logit(alpha[ii] .* (theta[jj] - beta[ii])); // Eq. 2.1
}

generated quantities {                  // # calculate correlation matrix
  corr_matrix[2] Omega;
  Omega = multiply_lower_tri_self_transpose(L_Omega);
}"
```

Note. The basic *Rstan*-specification of the two-parameter logistic model is based on Furr (2016). This code was written under *Rstan* Version 2.14.1. It was tested for functionality under the most recent version (*Rstan* 2.17.3; Stan Development Team, 2018). Equations refer to the equations in the main document.

References

Furr, D. C. (2016). Hierarchical two-parameter logistic item response model. Retrieved from http://mc-stan.org/users/documentation/case-studies/hierarchical_2pl.html.

Stan Development Team (2018). *Rstan: The R interface to Stan, version 2.17.3*. Retrieved from https://github.com/stan-dev/rstan/wiki/RStan-Getting-Started.